# LOAN/SAVE

### **Display**



(HP-41CX, Hewlett Packard 1983 and DM41X, SwissMicros 2020)

#### Overview<sup>1</sup>

Program L/S (with subprograms LOAN and SAVE) calculates monthly payments or savings based on an amount, an interest value and a period. The loan and savings program was written to review different ways of managing the mortgage and savings. Most banks can provide a nice listing but what if the interest changes? Below program may be of help. A feeling of nostalgia will be experienced when calculating a mortgage over a period of 30 years. The calculation loop has to be performed 360 times. The cursor has to move 30 times over the display which is shown by a year-counter.

#### **Calculations**

The loan on the basis of annuity is calculated by below algorithm in which **AMT** and **MP** represent the amount respectively monthly payment and **r** is the monthly interest and **m** the number of months:

$$MP = AMT \cdot \frac{r^m}{r^{m-1} + r^{m-2} + \ldots + r + 1}$$

$$MP = AMT \cdot \frac{r^m}{r \cdot (r \cdot (r \cdot (\ldots) + 1) + 1) + 1}$$

Savings calculation is based on below algorithm in which **SUM** and **MP** represent the lump sum respectively monthly payment:

$$SUM = r \cdot (r \cdot (r \cdot (...) + MP) + MP) + MP$$

in which r expresses the monthly interest derived from the annual interest I (in %) as follows:

$$r = 1 + \frac{I\%}{12 \cdot 100}$$

hooxies.nl 1 of 4 ©1999-2020 Auke Hoekstra

<sup>&</sup>lt;sup>1</sup> This program is copyright and is supplied without representation or warranty of any kind. The author assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof

## Example (1): LOAN

KEYSTROKES	DISPLAY	COMMENTS
		Run L/S (main entry)
[XEQ] [ALPHA] L/S [ALPHA]	L/57	Select L for loan
L <mark>[R/S]</mark>	AMI=7	Enter the loan AMOUNT
50000 <mark>[R/S]</mark>	I%:7	Enter the annual interest
7,5 <mark>[R/S]</mark>	N R.M <u>-</u> 7	Enter the duration in number of months
240 <mark>[R/S]</mark>	20	Count down in years
	19	
	•	
	{	
	M.P : 402,80	Monthly payment
[XEQ] [ALPHA] LOAN [ALPHA]	AMI <u>-</u> 7	Run LOAN directly
12000 <mark>[R/S]</mark>	I% = 7	Enter the annual interest
2 <mark>[R/S]</mark>	N R.M <u>-</u> 7	Enter the duration in number of months
36 <mark>[R/S]</mark>	3	Count down in years
	2	
	<b>{</b>	
	M.P = 343,71	Monthly payment
[R/S]	L/57	Run again from main entry

## Example (2): SAVE

KEYSTROKES	DISPLAY	COMMENTS
		Run L/S (main entry)
[XEQ] [ALPHA] L/S [ALPHA]	L/57	Select S for savings
S[R/S]	M.P <u>-</u> 7	Enter the monthly payment
150[R/S]	I% = 7	Enter the annual interest
3 <mark>[R/S]</mark>	N R.M = 7	Enter the duration in number of months
60[R/S]	5	Count down in years
	•	
	{	
	5UM:9.72 1,25	Lump sum savings
[CLX]	9.72 (,2494 (	Clear display
[XEQ] [ALPHA] SAVE [ALPHA]	M.P <u>-</u> 7	Run SAVE directly
999 <mark>[R/S]</mark>	I% = 7	Enter the annual interest
1,5[R/S]	N R.M = 7	Enter the duration in number of months
24 <mark>[R/S]</mark>	2	Count down in years
	<b>{</b>	
	5UM:24354,24	Lump sum savings
[R/S]	L/57	Run again from main entry

hooxies.nl 2 of 4 ©1999-2020 Auke Hoekstra

#### **Program Listing**

The LOAN program's outcome will be the Monthly Payment for the loan (mortgage) on the basis of annuity and requires the following parameters:

- the Amount to borrow in your own currency
- the effective Annual Interest
- the Number of Months to pay it back

The SAVE program calculates the final Lump Sum you receive after the period of saving and requires the following parameters:

- the Monthly Payment in your own currency
- the effective Annual Interest
- the Number of Months to save your money

The listing below shows an extended command in line 08: ATOX. This is to check whether the input character was an 'L' (value 76). For the C and CV this can be replaced by changing it an ASTO X followed by 'L'. The same applies to line 11 where the compare character is an 'S' (value 83).

The number of bytes can easily be reduced by removing subroutine LBL 06 and its calls XEQ 06 or by deleting lines 01-13.

The listing of program L/S (with subprograms LOAN and SAVE) is given below:

01•LBL "L/S"	22•LBL 02	43 CF 29	64 ,
02•LBL 00	23 XEQ 06	44 RTN	65∎LBL 03
03 "L/S?"	24 RCL 01	45∎LBL 06	66 XEQ 06
04 AON	25 +	46 RCL 03	67 RCL 02
05 PROMPT	26 RCL 02	47 STO 04	68 *
06 AOFF	27 *	48 RDN	69 1
07 76	28 DSE 03	49 12	70 +
08 ATOX	29 GTO 02	50 ST/ 04	71 DSE 03
09 X=Y?	30 "SUM="	51 X<> 04	72 GTO 03
10 GTO "LOAN"	31 GTO 07	52 FRC	73 /
11 83	32 <b>■</b> LBL 05	53 X=0?	74 *
12 X#Y?	33 PROMPT	54 VIEW L	75 "M.P="
13 GTO 00	34 STO 01	55 RDN	76∎LBL 07
14•LBL "SAVE"	35 "I%=?"	56 RTN	77 FIX 2
15 "M.P=?"	36 PROMPT	57 <b>■</b> LBL "LOAN"	78 ARCL X
16 XEQ 05	37 1200	58 "AMT=?"	79 FIX 5
17 *	38 /	59 XEQ 05	80 SF 29
18 "NR.M=?"	39 1	60 "NR.M=?"	81 AVIEW
19 PROMPT	40 +	61 PROMPT	82 END
20 STO 03	41 STO 02	62 STO 03	
21 CLX	42 FIX 0	63 Y^X	(177 bytes)

hooxies.nl 3 of 4 ©1999-2020 Auke Hoekstra

## Registers, Labels and Flags

REGISTERS	COMMENTS
R01	Loan amount or monthly payment
R02	Annual interest
R03	Duration in number of months
R04	Work register for count down

LABELS	COMMENTS	
LBL00	Main entry	
LBL02	Loop around savings	
LBL03	Loop around loan	
LBL05	Common stop for interest	
LBL06	Calculate year count down	
LBL07	Reset default, show result	

FLAGS	COMMENTS
29	Type of thousands separator

### **Downloads**

The RAW/TXT format of the program is available via the website: <u>LOAN+SAVE</u> (in zip file).

hooxies.nl 4 of 4 ©1999-2020 Auke Hoekstra